

Unpaarige Anführungszeichen in einem WORD-Dokument markieren

Inhalt

1	Aufgabe	2
2	Lösung	2
2.1	Funktionsprozedur	2
2.2	Subprozedur	3

Abbildungen

Abbildung 1:	Länderkürzel interaktiv bestimmen	3
Abbildung 2:	Absätze mit unpaarigen Anführungszeichen gelb markieren	4

Listings

Listing 1:	Länderkürzel interaktiv bestimmen	2
Listing 2:	Absätze mit unpaarigen Anführungszeichen gelb markieren	4

Literatur

- [1] A. Wyatt, „Checking for Missing Quotation Marks (Microsoft Word)“ 25 05 2020. [Online]. Available: https://wordribbon.tips.net/T013293_Checking_for_Missing_Quotation_Marks.html. [Zugriff am 15 03 2021].

1 Aufgabe

Zwei VBA¹-Prozeduren werden im Folgenden vorgestellt, die unpaarige Anführungszeichen in einem WORD-Dokument markieren:

- Die VBA-Subprozedur mit dem Namen „*MarkUnevenQuotes*“ markiert alle Absätze eines WORD-Dokuments, die unpaarige Anführungszeichen enthalten, egal ob dort *gerade* (sog. Gänsefüßchen) oder *doppelte typografische* Anführungszeichen vorkommen.²
- Die VBA-Funktionsprozedur mit dem Namen „*fctGetQuoteForm*“ dient zur interaktiven Auswahl von doppelten typografischen Anführungszeichen wie sie in Deutschland, England oder Frankreich verwendet werden.

Je nach Form der Anführungszeichen in den drei Sprachbereichen wird ...

- bei Verwendung sog. Gänsefüßchen geprüft, ob davon eine gerade Anzahl im betreffenden WORD-Dokument vorkommt.
- bei doppelten typografischen Anführungszeichen geprüft, ob für jedes öffnende das dazugehörige schließende Anführungszeichen vorhanden ist.

Wenn eine dieser beiden Bedingungen ungültig ist, wird die Fundstelle im betreffenden Absatz gelb hervorgehoben.

2 Lösung

2.1 Funktionsprozedur

Die VBA-Funktionsprozedur mit dem Namen „*fctGetQuoteForm*“ wird aufgerufen von der VBA-Subprozedur „*MarkUnevenQuotes*“. Den entsprechenden Quellcode enthält (s. Listing 1):

```
1 Function fctGetQuoteForm() As String
2     ' Aufgabe: Länderkürzel interaktiv bestimmen.
3     ' Aufruf von: VBA-Subprozedur "MarkUnevenQuotes"
4     Dim varQuotes As Variant ' Array der Länderkürzel: d, e, f
5     Dim strQuotes As String
6     Dim strPrompt As String
7     Dim strShort As String
8     Dim strQt As String * 1
9     Dim intIdx As Integer
10    varQuotes = Array("d", "e", "f") ' Kürzel für drei Länder
11    strPrompt = "Bitte Kürzel bestätigen oder ändern!" & vbCrLf
12    For intIdx = LBound(varQuotes) To UBound(varQuotes)
13        strShort = varQuotes(intIdx)
14        Select Case strShort
15            Case "d"
16                strPrompt = strPrompt & "deutsch:" & Space(5) & strShort & vbCrLf
17            Case "e"
18                strPrompt = strPrompt & "englisch:" & Space(5) & strShort & vbCrLf
19            Case "f"
20                strPrompt = strPrompt & "französisch:" & Space(1) & strShort & vbCrLf
21        End Select
22    Next intIdx
23    strQt = InputBox(strPrompt, "Form der doppelten Anführungszeichen", "d")
24    If Len(strQt) = 0 Then
25        MsgBox "Abbruch durch den Benutzer!", vbExclamation
26        fctGetQuoteForm = ""
27    Else
28        ' MsgBox strQt
29        fctGetQuoteForm = strQt
30    End If
31 End Function
```

Listing 1:Länderkürzel interaktiv bestimmen

¹ VBA ist ein Akronym für **V**isual **B**asic for **A**pplications

² In Anlehnung an [1].

Abbildung 1 demonstriert die interaktive Auswahl des Länderkürzels (d, e oder f):

Abbildung 1: Länderkürzel interaktiv bestimmen

In Abhängigkeit vom gewählten Länderkürzel (beispielsweise „d“, siehe oben) wird im Quellcode (siehe Listing 2, Codezeilen 14, 15 und 16) das öffnende bzw. schließende Anführungszeichen bestimmt.

2.2 Subprozedur

Die VBA-Subprozedur mit dem Namen „MarkUnevenQuotes“ beinhaltet folgenden Quellcode (s. Listing 2):

```

1 Sub MarkUnevenQuotes()
2 ' Aufgabe: Absätze mit unpaarigen Anführungszeichen gelb markieren
3 Dim strText As String, strShortNm As String
4 Dim strOpening As String * 1, strClosing As String * 1
5 Dim intNormal, intSmart As Integer
6 Dim lngCnt As Long
7 On Error GoTo ErrHandler
8 strShortNm = fctGetQuoteForm ' VBA-Funktionsprozedur aufrufen
9 If strShortNm = vbNullString Then
10 MsgBox "Abbruch: Ungültiges Kürzel!", vbCritical
11 GoTo ExitPoint
12 End If
13 Select Case strShortNm
14 Case "d" ' -> deutsch: doppelt
15 strOpening = Chr(132) ' öffnend
16 strClosing = Chr(147) ' schließend
17 Case "e" ' -> englisch: doppelt
18 strOpening = Chr(147) ' öffnend
19 strClosing = Chr(148) ' schließend
20 Case "f" ' -> französisch: doppelt
21 strOpening = Chr(171) ' öffnend
22 strClosing = Chr(187) ' schließend
23 Case Else
24 MsgBox "Abbruch: Laufzeitfehler!", vbExclamation
25 GoTo ExitPoint
26 End Select
27 Application.ScreenUpdating = False
28 ActiveDocument.Select
29 With Selection
30 .Range.HighlightColorIndex = wdNoHighlight ' Gelbe Markierungen entfernen
31 .HomeKey Unit:=wdStory
32 .Find.ClearFormatting
33 With .Find
34 .Text = Chr(34)
35 .Replacement.Text = ""
36 .Forward = True
37 .Wrap = wdFindStop
38 .Format = True
39 .MatchCase = False
40 .MatchWholeWord = False

```

```

41         .MatchWildcards = False
42         .MatchSoundsLike = False
43         .MatchAllWordForms = False
44         .Execute
45     End With
46     Do While .Find.Found
47         .MoveDown Unit:=wdParagraph, Count:=1, Extend:=wdExtend
48         .MoveLeft Unit:=wdCharacter, Count:=1, Extend:=wdExtend
49         strText = .Text
50         intNormal = 0 'Zähler initiieren f. gerade (d. h. normale) Anführungszeichen
51         intSmart = 0 'Zähler initiieren f. typografische Anführungszeichen
52         For lngCnt = 1 To Len(strText)
53             Select Case Mid(strText, lngCnt, 1)
54                 Case Chr(34) 'gerades (d.h. normales) Anführungszeichen
55                     If intNormal > 0 Then
56                         intNormal = intNormal - 1
57                     Else
58                         intNormal = intNormal + 1
59                     End If
60                 Case strOpening 'öffnend
61                     intSmart = intSmart + 1
62                 Case strClosing 'schließend
63                     intSmart = intSmart - 1
64             End Select
65         Next lngCnt
66         If intNormal <> 0 Or intSmart <> 0 Then
67             .Range.HighlightColorIndex = wdYellow 'Unpaarigkeit gelb markieren
68         End If
69         .Collapse Direction:=wdCollapseEnd
70         .Find.Execute
71     Loop 'Ende der Do-While-Schleife
72     .HomeKey Unit:=wdStory
73 End With
74 ExitPoint:
75 On Error Resume Next
76 Application.ScreenUpdating = True
77 MsgBox "Fertig!", vbExclamation
78 Exit Sub
79 ErrHandler:
80 MsgBox Err.Number & " Laufzeitfehler: " & Err.Description
81 Resume ExitPoint
82 End Sub

```

Listing 2: Absätze mit unpaarigen Anführungszeichen gelb markieren

Mit der RAND()-Funktion³ zufällig erzeugter Beispielttext (sog. Blindtext) mit nachträglich eingebauten unpaarigen Anführungszeichen zum Testen der vorstehenden Prozeduren (s. Abbildung 2).

Video bietet eine leistungsstarke Möglichkeit zur Unterstützung Ihres Standpunkts. Wenn Sie auf „Onlinevideo klicken, können Sie den Einbettungscode für das Video einfügen, das hinzugefügt werden soll. Sie können auch ein Stichwort eingeben, um online nach dem Videoclip zu suchen, der optimal zu Ihrem Dokument passt.

Damit Ihr Dokument ein professionelles Aussehen erhält, stellt Word einander ergänzende Designs für Kopfzeile, Fußzeile, Deckblatt und Textfelder zur Verfügung. Beispielsweise können Sie ein passendes Deckblatt mit Kopfzeile und Randleiste hinzufügen. Klicken Sie auf Einfügen² und wählen Sie dann die gewünschten Elemente aus den verschiedenen Katalogen aus.

Designs und Formatvorlagen helfen auch dabei, die Elemente Ihres Dokuments aufeinander abzustimmen. Wenn Sie auf „Entwurf klicken und ein neues Design auswählen, ändern sich die Grafiken, Diagramme und SmartArt-Grafiken so, dass sie dem neuen Design entsprechen. Wenn Sie Formatvorlagen anwenden, ändern sich die Überschriften passend zum neuen Design.

Video bietet eine leistungsstarke Möglichkeit zur Unterstützung Ihres **Standpunkts**.

Abbildung 2: Absätze mit unpaarigen Anführungszeichen gelb markieren

³ =RAND (Anzahl Absätze, Anzahl Sätze) Funktion in MS WORD.